

大学の一般教育における「数学」への 電子計算機活用の試み

斎藤 瞭*・森本 仁**

The Practical Use of a Personal Computer in Mathematics in General Education at College Level

Ryo SAITO* and Masashi MORIMOTO**

(September, 1989)

はじめに

大学の一般教育における数学の目的は「専門教育のための基礎としての数学」から「教養としての数学」まで多様である。一方、高等学校においては昭和57年度入学から指導要領が改訂され、新指導要領によって学んだ学生が昭和60年度から大学に入学してくるようになった。新指導要領の特色は[2]に述べられているように、大学との接続のことを考慮するというよりは、もし多様な履修状況の学生が入学しているならば、大学との接続についてはそれぞれの大学が工夫していくという考え方である。

我々の経験によれば、近年の大学生は高等学校段階の履修状況の多様性のみならず、内容の理解度の多様性も併せ持っている。従って大学の一般教育においても高等学校との接続としての教育内容とともに、教育方法も考慮される必要がある。([4] に中学1年段階で数学について成績と指導方法との相関関係は認められなかったとの報告がある。しかし、だから、だからといって教育方法はどうしてもよいということにはならないであろう。米国マサチューセッツ工科大学には教師の工夫を求めたかなり細かいガイドラインがある[6]。)

そこで一般教育の数学の理解のためにコンピュータを活用することを考えてみる。活用方法もいろいろあって、主として成績処理等の統計・事務処理に用いるもの

(CMI=computer managed instruction), 教師の代行的な役割を持たせるもの(CAI=computer aid instruction…大学の一般教育への試みとしては[3]がある。)もあるが、ここでは数値計算やグラフの作成に応用することを試みることにしたい。この目的は教師や学生が具体的にやってみるということが困難な部分をコンピュータを使用することによって、学生の数学理解の一助にしたいということである。ただし、コンピュータ教室で数学の授業を行うことを目標にしているのではなく、具体的には教科書の中で展開することにしたのである。またもし時間的な余裕があれば、プログラムの本質的な部分を解説することも数学理解にとって有益であろう。

このような考え方の高等学校における数学への試みとしては[5]がある。我々の試みはその延長にあると言ってもよい。我々はコンピュータを活用するにあたり、次の2点に留意した。

- (1) ハード面、ソフト面での計算上の厳密性はそれほど要求せず、パソコンのDISK BASICでやれる程度の範囲とする。(数値のさらにある程度の厳密さを要求するとすれば[1]がある。)
- (2) なるべく簡単な構造のプログラムを目指す。できるならば学生が自主的に実行可能な程度のものがよい。

具体的な事項としては以下のことについて試みた。

§1. 準備

* 酪農科, 数学研究室 斎藤 瞭

Department of Dairy Science (Mathematics), Hokkaido College of Arts and Sciences Ebetsu, Hokkaido 069, Japan

** 教養科, 数学研究室 森本 仁

Department of General Education (Mathematics), Rakuno Gakuen University Ebetsu, Hokkaido 069, Japan

- § 2. 極限の概念
- § 3. 連続関数の実根
- § 4. 行列・行列式

§ 1. 準 備

(1) 命 題

命題とその対偶の関係を強調する以外に特に命題を扱う必要はないかもしれないが、条件分岐 IF ~ THEN ~ ELSE ~ の教育には効果を期待できる。たとえば、

```
500 *DISPCONDITION
510 FOR P=-1 TO 0 : FOR Q=-1 TO 0
530     IF (NOT(P) OR Q)=-1 THEN PRINT "   P=";AKCNV$(RIGHT$(STR$(P),1));
                                   "   Q=";AKCNV$(RIGHT$(STR$(Q),1))
540     NEXT Q : NEXT P
570 RETURN
```

により

```
RUN
P=>Q が真の場合はつぎの通り
  P=1   Q=1
  P=0   Q=1
  P=0   Q=0
Ok
```

などはよい例と思う。

(2) 最大公約数

BASIC プログラムを実行させてみて、2数の場合と3数以上の場合の取扱い、あるいは0を含む場合も考慮する必要があることが判った。

2数の場合

```
440 *GCMSUB1
450 IF AA=0 OR BB=0 THEN GCM=1 : RETURN
460 IF AA<BB THEN SWAP AA,BB
470 *LOOP : RR=AA-(AA ¥ BB)*BB
480 IF RR=0 THEN GCM=BB : RETURN
490 IF RR<>0 THEN AA=BB:BB=RR : GOTO *LOOP
500 RETURN
```

により

```
RUN
      Old Set = (-1734 , 2652 ) ==> New Set = (-17 , 26 )
GCM   = 102           the time needed = 0
```

で、1秒以内に完了する。

3数以上の場合

```
440 *GETGCM
450 FOR J=1 TO N
460     AA=GCM:BB=AR(J)
470     IF AA<=BB THEN SWAP AA,BB
480     IF BB=0 THEN GCM=AA: GOTO *BRK
490     *LOOP : RR=AA-(AA ¥ BB)*BB
```

```

500     IF RR=0 THEN GCM=BB : GOTO *BRK
510     IF RR<>0 THEN AA=BB:BB=RR : GOTO *LOOP
520     *BRK
530     NEXT J
540     RETURN

```

により

```

RUN
Old Set = ( 1239 , -777 , 1281 , -483 )
==> New Set = ( 59 , -37 , 61 , -23 )
GCM    = 21
the time needed = 0 sec
Ok

```

で1秒以内で終わる。なお、3数以上の最大公約数は配列変数を用いるのが簡単である。

(3) 順 列

n 個の文字の順列は配列変数を使うが、ここでは簡単にするために4文字の順列を求めてみる。

```

460 *PERMUTATION
470   FOR I=1 TO N : FOR J=1 TO N
490     IF J=I THEN GOTO *BRKJ
500     FOR K=1 TO N
510       IF K=I OR K=J THEN GOTO *BRKK
520       FOR L=1 TO N
530         IF L=I OR L=J OR L=K THEN GOTO *BRKL
540         IF SG*(L-K)*(L-J)*(L-I)*(K-J)*(K-I)*(J-I)>0 THEN CN=CN+1 :
           PRINT STR$(I);STR$(J);STR$(K);STR$(L);" ";
550       *BRKL : NEXT L
560     *BRKK : NEXT K
570     IF (CN MOD 6)=0 THEN PRINT
580   *BRKJ : NEXT J : NEXT I
600   RETURN

```

を実行すると

```
RUN
```

偶順列は

```

1 2 3 4   1 3 4 2   1 4 2 3   2 1 4 3   2 3 1 4   2 4 3 1
3 1 2 4   3 2 4 1   3 4 1 2   4 1 3 2   4 2 1 3   4 3 2 1

```

奇順列は

```

1 2 4 3   1 3 2 4   1 4 3 2   2 1 3 4   2 3 4 1   2 4 1 3
3 1 4 2   3 2 1 4   3 4 2 1   4 1 2 3   4 2 3 1   4 3 1 2

```

```
the time needed = 2
```

(4) 方程式のグラフ

方程式のグラフを

1) 媒介変数表示

$$x = \phi(t), y = \psi(t) \quad (a \leq t \leq b)$$

2) 陰関数表示

$$F(x, y) = 0$$

の2つの場合に分けて考える。

例 関数 $y=f(x)$ ($a \leq x \leq b$) は

$$\begin{cases} x=t \\ y=f(t) \end{cases} \quad (a \leq t \leq b)$$

関数 $r=f(\theta)$ ($a \leq \theta \leq b$) は

$$\begin{cases} x=f(\theta) \cos \theta \\ y=f(\theta) \sin \theta \end{cases} \quad (a \leq \theta \leq b)$$

として扱うことができる。陰関数 $x^2-xy+y^2=1$ でも

$$\begin{cases} x=(2/\sqrt{3}) \cos \theta \\ y=(1/\sqrt{3}) \cos \theta + \sin \theta \end{cases} \quad (0 \leq \theta \leq 2\pi)$$

とすれば、媒介変数表示の場合に帰着できることがある。

1) の場合は

```
100 RS=RATE_SPREAD : A=X_ORIGIN : B=Y_ORIGIN : SP=STEP_WIDTHH
110 DEF FNF(T)=F(T) : DEF FNG(T)=G(T)
120 FOR T=M TO N STEP SP
130   X=A+RS*FNF(T) : Y=B-RS*FNG(T) : PSET (X,Y),6
150 NEXT T
```

2) の場合は

```
100 RS=RATE_SPREAD : A=X_ORIGIN : B=Y_ORIGIN : SP=STEP_WIDTHH
110 DEF FNF(X,Y)=F(X,Y)
120 FOR X=M TO N STEP SP : FOR Y=P TO Q STEP SP
140   FR=FNF(X,Y) : FX=FNF(X+SP,Y) : FY=FNF(X,Y+SP)
150   IF FR*FY<0 THEN PSET (RS*X+A,B-RS*(Y-FR*SP/(FY-FR))),6
160   IF FR*FX<0 THEN PSET (A+RS*(X-FR*SP/(FX-FR)),-RS*Y+B),6
170 NEXT Y : NEXT X
```

によってそのグラフを描くことができる。

平行移動・対称移動などについては、移動前と移動後を色を変えて表示すれば視覚的に効果を期待できる。たとえば、グラフの直線 $y=x$ に関する対称移動については

```
100 RS=RATE_SPREAD : A=X_ORIGIN : B=Y_ORIGIN : SP=STEP_WIDTHH
110 DEF FNF(T)=F(T) : DEF FNG(T)=G(T)
120 FOR T=M TO N STEP SP
130   X1=A+RS*FNF(T) : Y1=B-RS*FNG(T) : PSET (X1,Y1),6
140   Y2=A+RS*FNG(T) : X2=B-RS*FNF(T) : PSET (Y2,X2),5
150 NEXT T
```

によってそのグラフを描けばよい。

xy-座標を、細部については少しの手直しが必要であるが、統一して

```
350 *DISPLANE
360 LINE (A-2.5*RS,B)-(A+3.5*RS,B) : ' x軸の設定
370 LINE (A+3.5*RS,B)-STEP(-4,-4) : LINE (A+3.5*RS,B)-STEP(-4,4)
400 LINE (A,B-2*RS)-(A,B+RS) : ' y軸の設定
410 LINE (A,B-2*RS)-STEP(-4,4) : LINE (A,B-2*RS)-STEP(4,4)
440 CIRCLE (A-10,B+10), 5,,,1.5 : ' 原点0を書き込む
450 RETURN
```

により描くこととすればよい。

例 $r^2=\cos(2\theta)$

```

330 *DISPCURVE
340 DEF FNF(T)=COS(T)/(1+SIN(T)*SIN(T))
350 DEF FNG(T)=SIN(T)*COS(T)/(1+SIN(T)*SIN(T))
360 FOR T=0 TO 2*PI STEP SP
370   X=A+RS*FNF(T) : Y=B-RS*FNG(T) : PSET (X,Y),6
390 NEXT T
430 RETURN

```

例 $x^2 - xy + y^2 = 1$ のグラフ

```

390 *DISPCURVE1
400 DEF FNF(X,Y)=X^2-X*Y+Y^2-1
410 FOR X=-1.16 TO 1.16 STEP SP : FOR Y=-1.16 TO 1.16 STEP SP
430   FR=FNF(X,Y) : FY=FNF(X,Y+SP) : FX=FNF(X+SP,Y)
440   IF FR*FY<0 THEN PSET (A+RS*X,B-RS*(Y-FR*SP/(FY-FR))),6
450   IF FR*FX<0 THEN PSET (A+RS*(X-FR*SP/(FX-FR)),B-RS*Y),6
460 NEXT Y : NEXT X
480 RETURN
490 '
500 *DISPCURVE2
510 DEF FNF(T)=(2/SQR(3))*COS(T) : DEF FNG(T)=(1/SQR(3))*COS(T)+SIN(T)
530 FOR T=0 TO 2*PI STEP .5*SP
540   X=A+RS*FNF(T) : Y=B-RS*FNG(T) : PSET (X,Y),6
560 NEXT T
570 RETURN

```

§2. 極限の概念

パソコンは繰り返し実行するような数値計算に効果がある。このことを極限の概念を理解させる方法に応用することができる。

(1) 数列の極限

例1. $\lim_{n \rightarrow \infty} \sqrt[n]{n} = 1$

```

100 AN=N^(1/N)
110 PRINT AN

```

実行結果	
N = 100	1.047128558158875
N = 500	1.012506723403931
N = 1000	1.006931662559509
N = 5000	1.001704931259155
N = 10000	1.000921487808228

例1について $\sqrt[n]{n}$ の近似値を Newton 法によって計算してみたのが次の結果である。

$f(x) = x^n - n$, $f'(x) = nx^{n-1}$, $x_0 = 2$, $x_j = x_{j-1} - \frac{f(x_{j-1})}{f'(x_{j-1})}$ ($j=1, 2, 3, \dots$) としている。

```

100 DIM X(100)

```

```

110 X(0)=2
120 FOR J=1 TO 100
130   X(J)=X(J-1)-(X(J-1)^N-N)/(N*X(J-1)^(N-1))
140 NEXT J
150 PRINT X(100)

```

実行結果	
N = 1	1
N = 5	1.379729661461215
N = 10	1.258925411794167
N = 50	1.081382656800293
N = 100	1.0471285480509
N = 110	1.043657780939918
N = 120	1.040702290439091

この場合、 $X(0)$ の値によって計算できる N の値が異なる。(例えば $X(0)=10$ とすると $N=37$)。なお N が大きい値 (例えば $N=100$) となると、 $J=80$ 程度としないと $\sqrt[n]{n}$ のよい値が得られない。

$$\text{例 2. } \lim_{n \rightarrow \infty} \left(2 + \frac{1}{2!} + \frac{1}{3!} + \cdots + \frac{1}{n!} \right) = e$$

階乗の計算を含む場合、我々は $33!$ までしか計算できない。この例では、それでもよい近似値が得られている。

	実行結果	誤 差
100 FOR N=1 TO 33 STEP 5		
110 S=1	N = 3 2.666666666666667	(< 10^{-1})
120 FOR J=1 TO N	N = 8 2.71827876984127	(< 10^{-5})
130 GOSUB *FACTORIAL	N = 13 2.718281828446759	(< 10^{-10})
140 S=S+1/T	N = 18 2.718281828459045	(< 10^{-15})
150 NEXT J	N = 23 2.718281828459045	(< 10^{-15})
160 PRINT S	N = 28 2.718281828459045	(< 10^{-15})
170 NEXT N	N = 33 2.718281828459045	(< 10^{-15})
180 '		
190 END		
200 '		
210 *FACTORIAL		
220 T=1		
230 FOR K=1 TO J		
240 T=T*K		
250 NEXT K		
260 RETURN		

$$\text{例 3. } \lim_{n \rightarrow \infty} \left(1 + \frac{1}{2} + \frac{1}{3} + \cdots + \frac{1}{n} \right) = \infty$$

	実行結果
100 S=0	
110 FOR K=1 TO N	N = 10 2.928968253968254
120 S= S+1/K	N = 50 4.499205338329425
130 NEXT K	N = 100 5.18737751763962
140 PRINT S	N = 500 6.792823429990524
	N = 1000 7.485470860550344
	N = 5000 9.094508852984434

この実行結果では、この数列が無限大に発散することを直接みることはできないが、次の例によって発散する数列 $\{\log n\}$ との比較で調べることができる。

$$\text{例 4. } \lim_{n \rightarrow \infty} \frac{1 + \frac{1}{2} + \frac{1}{3} + \cdots + \frac{1}{n}}{\log n} = 1$$

	実行結果
100 S=0	
110 FOR K=1 TO N	N = 10 1.272034732703779
120 S= S+1/K	N = 50 1.150096902531462
130 NXET K	N = 100 1.126424700087553
140 T= S/LOG(N)	N = 500 1.093041301984204
150 PRINT T	N = 1000 1.083632881346722
	N = 5000 1.067782428226576
	N = 10000 1.06267580836714
	N = 50000 1.053349105988582

(注) 例 3 との関係で $\lim_{n \rightarrow \infty} \left(1 - \frac{1}{2} + \frac{1}{3} - \frac{1}{4} + \cdots + (-1)^{n-1} \cdot \frac{1}{n} \right) = \log 2$ の収束状況をみておく。
 $(\log 2 \doteq 0.693147180559945 \cdots)$

誤 差

N = 10	.6456349206349206	(< 10 ⁻¹)
N = 50	.6832471605759182	(< 10 ⁻²)
N = 100	.6881721793101952	(< 10 ⁻²)
N = 500	.6921481805579452	(< 10 ⁻³)
N = 1000	.6926474305598202	(< 10 ⁻³)
N = 5000	.6930471905599449	(< 10 ⁻⁴)

例 5. $\lim_{n \rightarrow \infty} \sqrt[n]{n!} = \infty$

これについても例 8 と同様に $\sqrt[n]{n!}$ が無限大に発散することをみることはできない。

従って $\lim_{n \rightarrow \infty} \frac{n}{\sqrt[n]{n!}} = e$ を確認する。なお $\frac{n}{\sqrt[n]{n!}} = \sqrt[n]{\frac{n}{n-1}} \cdot \sqrt[n]{\frac{n}{n-2}} \cdots \sqrt[n]{\frac{n}{n-(n-1)}}$ として計算する。

100 T=1	実行結果	誤 差
110 FOR K=1 TO N-1	N = 10	2.20812507272605 (< 10 ⁻⁰)
120 T= T*(N/(N-K))^(1/N)	N = 50	2.56630687549675 (< 10 ⁻⁰)
130 NEXT K	N = 100	2.632085655283513 (< 10 ⁻¹)
140 PRINT T	N = 500	2.69648227877727 (< 10 ⁻¹)
	N = 1000	2.706421251871661 (< 10 ⁻¹)
	N = 5000	2.715465072021249 (< 10 ⁻²)
	N = 10000	2.716770786445022 (< 10 ⁻²)

(2) 収束する数列に対し、いわゆる ε - n_0 論法により、 ε を与えたときの自然数 n_0 を求めること。

これは収束の定義に係わる概念であるが、初学者にとってはかなり難解といえるものである。従って具体的な数列で ε より n_0 を示してやるのがよいと思われる。我々は無限数列は扱えないので、有限数列で例を考えることは止むを得ない。

例 6. $a_n = \frac{1}{1+(n-100)^2} + 5$ のとき $\lim_{n \rightarrow \infty} a_n = 5$

この数列に対して E を与えたとき、 $M < n$ であるような n に対して常に $|a_n - 5| < E$ となる自然数 M を求める。

100 DIM A(200)	実行結果
110 FOR N=1 TO 200	E=1 M=100
120 A(N)=1/(1+(N-100)^2)+5	E=0.1 M=103
130 NEXT N	E=0.01 M=109
140 INPUT E	E=0.001 M=131
150 IF A(200)<5-E OR 5+E<A(200) THEN GOTO *EXIT	E=0.0001
160 K=1:GOTO *SEARCH1	(a ₂₀₀ までには M が求まらない)
170 '	
180 *EXIT	
190 END	
200 '	
210 *SEARCH1	
220 FOR J=K TO 200	
230 IF 5-E<A(J) AND A(J)<5+E THEN M=J:GOSUB *SEARCH2	
240 IF FLAG=1 THEN PRINT "e";E,"M=";M-1;:GOTO *EXIT	
250 NEXT J	
260 RETURN	
270 '	

```
280 *SEARCH2
290 FOR P= M+1 TO 200
300   IF A(P)<=5-E OR 5+E<=A(P) THEN K=P:GOTO *SEARCH1
310 NEXT P
320 FLAG=1
330 RETURN
```

一般に $\lim_{n \rightarrow \infty} a_n = a$ ならば $\lim_{n \rightarrow \infty} \frac{a_1 + a_2 + \dots + a_n}{n} = a$ が成立つ。この証明は通常いわゆる「 ϵ - n_0 論法」によってなされ、 $\lim_{n \rightarrow \infty} \frac{2n^2 + n + 1}{n^2 + n} = 2$ などと同様に直感的に理解させることは無理と思われる。そこで数値計算とともに ϵ を与えて n_0 を求めることが意味をもつ。

例 7. $\lim_{n \rightarrow \infty} \frac{1 + \frac{1}{2} + \dots + \frac{1}{n}}{n} = 0$

例 4 において $\log n$ の代りに n とおいた数値計算による実行結果

N = 10	.2928968253968251
N = 50	.0899841067665885
N = 100	.0518737751763962
N = 500	.01358564685998105
N = 1000	7.485470860550344D-03
N = 5000	1.818901770596887D-03
N = 10000	9.787606036044386D-04
N = 50000	2.279400789855693D-04

ϵ を与えて n_0 を求めることは例 4 と例 6 の program を組み合わせればよい。実行結果は $\epsilon = 0.1$ のとき $n_0 = 43$, $\epsilon = 0.07$ のとき $n_0 = 68$, $\epsilon = 0.05$ のとき $n_0 = 104$, $\epsilon = 0.03$ のとき $n_0 = 195$

(3) 関数の極限值, 微分係数


関数の極限值については, グラフと数値計算の両方から確認するのがよいと思われる。

例 8. $\lim_{x \rightarrow 0} \frac{\sin x}{x} = 1$

(グラフ)

実行結果

```
100 A=300:B=200:S=25
110 PAI= 3.141592
120 FOR T=-3*PAI TO 3*PAI STEP .01
130   IF -.001<T AND T<.001 THEN T=.001
140   X=S*T+A
150   Y=-S*1/T*SIN(T)+B
160   PSET (X,Y)
170 NEXT T
```



(数値計算)

実行結果

100 X= 1#/K	X = 1	(K = 1)	.8414709848078964
110 Y= SIN(X)/X	X = .5	(K = 2)	.9588510772084059
120 PRINT Y	X = .1	(K = 10)	.9983341664682814
	X = .05	(K = 20)	.9995833854135664
	X = .01	(K = 100)	.9999833334166663
	X = .005	(K = 200)	.9999958333385415

例 9. $f(x) = \frac{1}{1+e^x} \cdot x e^x$ ($x \neq 0$), $f(0) = 0$ の関数の $x = 0$ における左右微分係数


(グラフ)

```

100 A=300:B=200:S=50
110 FOR T=-3 TO 3 STEP .01
120   IF -.012<T AND T<.012 THEN T=.012
130   X=S*T+A
140   Y=-S*T*EXP(1/T)/(1+EXP(1/T))+B
150   PSET (X,Y)
160 NEXT T

```

実行結果



(数値計算)

```

100 DEF FNA(X)=X*EXP(1/X)/(1+EXP(1/X))
110 FOR N=1 TO 16 STEP 3
120   H=1#/N
130   PRINT FNA(-H)/-H,
140   PRINT FNA(H)/H
150 NEXT N

```

実行結果

左微分係数	右微分係数
.2689414213699951	.7310585786300049
.01798620996209156	.9820137900379084
9.110511944006454D-04	.9990889488055994
4.53978687024344D-05	.9999546021312976
2.260324297903575D-06	.9999977396757021
1.12535162055095D-07	.9999998874648379

(4) いわゆる「 ε - δ 論法」で ε を与えて δ を求めること。

例 10. $f(x) = \frac{1}{1+x^2}$ の $x=1$ における連続性

```

100 DEF FNA(X)=1/(1+X^2)
110 INPUT "イプシロン=";E
120 IF FNA(1+1/1000)<FNA(1)-E OR FNA(1)+E<FNA(1+1/1000) OR
    FNA(1-1/1000)<FNA(1)-E OR FNA(1)+E<FNA(1-1/1000)
    THEN GOTO *EXIT
130 K=1:GOTO *SEARCH1
140 '
150 *EXIT
160 END
170 '
180 *SEARCH1
190 FOR T=K TO 1000
200   IF FNA(1)-E<FNA(1+1/T) AND FNA(1+1/T)<FNA(1)+E
      AND FNA(1)-E<FNA(1-1/T) AND FNA(1-1/T)
      <FNA(1)+E THEN N=T:GOSUB *SEARCH2
210   IF FLAG=1 THEN PRINT "d=";1#/N:GOTO *EXIT
220 NEXT T
230 RETURN
240 '
250 *SEARCH2
260 FOR J=N+1 TO 1000
270   IF FNA(1)-E>FNA(1+1/J) OR FNA(1+1/J)>FNA(1)+E
      OR FNA(1)-E>FNA(1-1/J) OR FNA(1-1/J)>FNA(1)+E
      THEN K=J:GOTO *SEARCH1
280 NEXT J
290 FLAG=1
300 RETURN

```

実行結果

```

イプシロン=? 0.2
d = .3333333333333333
イプシロン=? 0.01
d = .0196078431372549
イプシロン=? 0.05
d = .09090909090909091
イプシロン=? 0.005
d = 9.900990099009901D-03

```

(5) 連続関数において $x_n \rightarrow a$ のとき $f(x_n) \rightarrow f(a)$ をみること。

例 11. $f(x) = \frac{1}{1+x^2} \quad (a=1)$

```
100 DEF FNA(X)=1/(1+X^2)
110 PRINT ABS(FNA(1+1/T)-FNA(1)),
120 PRINT ABS(FNA(1-1/T)-FNA(1))
```

実行結果

T = 5			
1-1/ 5	.1097561120986938	1+1/ 5	.0901639461517334
T = 10			
1-1/ 10	.05248618125915527	1+1/ 10	.04751130938529968
T = 50			
1-1/ 50	.01010000705718994	1+1/ 50	9.900033473968506D-03
T = 100			
1-1/ 100	5.024969577789307D-03	1+1/ 100	4.974991083145142D-03

(6) 積分の計算

シンプソンの公式，区分求積法，台形公式の比較をしてみる。

例 12. $\int_0^1 \sqrt{1-x^2} dx = \frac{\pi}{4}$

(シンプソンの公式)

```
100 DEF FNA(X)=SQR(1-X^2)
110 TU=0:TV=0
120 FOR J=2 TO 2*N-2 STEP 2
130 TU=TU+FNA(1#/(2*N)*J)
140 NEXT J
150 FOR J=1 TO 2*N-1 STEP 2
160 TV=TV+FNA(1#/(2*N)*J)
170 NEXT J
180 S=1#/(6*N)*(1+2*TU+4*TV)
190 PRINT S
```

(区分求積法 I)

```
100 DEF FNA(X)=SQR(1-X^2)
110 S=0
120 FOR J=0 TO N-1
130 S=S+FNA (1#/N*J)/N
140 NEXT J
150 PRINT S
```

(区分求積法 II)

```
100 DEF FNA(X)=SQR(1-X^2)
110 S=0
120 FOR J=1 TO N
130 S=S+FNA(1#/N*J)/N
140 NEXT J
150 PRINT S
```

(台形公式)

```
100 DEF FNA(X)=SQR(1-X^2)
110 S=0
120 FOR J=1 TO N
130 S=S+(FNA(1#/N*(J-1))+FNA(1#/N*J))/2*(1#/N)
140 NEXT J
150 PRINT S
```

実行結果

種類	[0,1]の分割	50 等 分	100 等 分	500 等 分
シンプソンの公式	N = 25	.785073143641154	N = 50 .7852833013733228	N = 250 .7853878915309906
区 分 求 積 法 I	N = 50	.7945671239867806	N = 100 .7901042546145618	N = 500 .7863718659646111
区 分 求 積 法 II	N = 50	.7745671244338155	N = 100 .7801042548380792	N = 500 .7843718658696162
台 形 公 式	N = 50	.7845671387016773	N = 100 .7851042704284191	N = 500 .7853718805536628

$\frac{\pi}{4} \doteq 0.785398163397748...$

(7) Maclaurin 展開式によるグラフ


$$\text{例 13. } \sin x \doteq x - \frac{x^3}{3!} + \frac{x^5}{5!} - \cdots + (-1)^n \cdot \frac{x^n}{(2n+1)!}$$

```

100 PAI= 3.14159
110 A= 300:B=200:S=50
120 INPUT "X ノ ジスウ 2N+1 , N ノ シテイ (N>=1) N=";N
130 DIM P(N):DIM Q(N)
140 GOSUB *FACTORIAL
150 GOSUB *GRAPHSIN
160 GOSUB *GRAPHX
170 GOSUB *GRAPH
180 '
190 END
200 '
210 *FACTORIAL
220 FOR J=1 TO N
230   Q=1
240   FOR K=1 TO 2*J+1
250     Q=Q*K
260   NEXT K
270   Q(J)=Q
280 NEXT J
290 RETURN
300 '
310 *GRAPHSIN
320 FOR T=-PAI TO PAI STEP .01
330   X=S*T+A
340   Y=-S*SIN(T)+B
350   PSET (X,Y)
360 NEXT T
370 RETURN
380 '
390 *GRAPHX
400 FOR T=-PAI TO PAI STEP .05
410   X=S*T+A
420   Y=-S*T+B
430   PSET (X,Y)
440 NEXT T
450 RETURN
460 '
470 *GRAPH
480 FOR M=1 TO N
490   FOR T=-PAI TO PAI STEP .05
500     P(0)=T

```

実行結果
N=3 の場合



```

510      FOR J=1 TO M
520          P(J)=P(J-1)+(-1)^J*(T^(2*J+1))/Q(J)
530      NEXT J
540      X=S*T+A
550      Y=-S*P(M)+B
560      PSET (X,Y)
570  NEXT T
580 NEXT M
590 RETURN

```

(8) Fourier 級数展開式によるグラフ

例 14. $\frac{x}{2} \doteq \sin x - \frac{\sin 2x}{2} + \frac{\sin 3x}{3} - \dots + (-1)^n \cdot \frac{\sin nx}{n} \quad (-\pi < x < \pi)$


```

100 PAI=3.14159
110 A=300:B=200:S=50
120 INPUT "N ノ シテイ (N)=1) N=";N
130 DIM P(N)
140 GOSUB *GRAPHX2
150 GOSUB *GRAPHSIN
160 GOSUB *GRAPH
170 '
180 END
190 '
200 *GRAPHX2
210 FOR T=-PAI TO PAI STEP .01
220     X=S*T+A
230     Y=-S*T/2+B
240     PSET (X,Y)
250 NEXT T
260 RETURN
270 '
280 *GRAPHSIN
290 FOR T=-PAI TO PAI STEP .05
300     X=S*T+A
310     Y=-S*SIN(T)+B
320     PSET (X,Y)
330 NEXT T
340 RETURN
350 '
360 *GRAPH
370 FOR M=2 TO N
380     FOR T=-PAI TO PAI STEP .05
390         P(1)=SIN(T)
400         FOR J=2 TO M
410             P(J)=P(J-1)+(-1)^(J-1)*SIN(J*T)/J

```

実行結果

N=3 の場合



```

420     NEXT J
430     X=S*T+A
440     Y=-S*P(M)+B
450     PSET (X,Y)
460     NEXT T
470 NEXT M
480 RETURN
    
```

§ 3. 連続係数の実根

$[a, b]$ に $f(x)=0$ の実根があることがわかっている場合, $f(x-0.001) \cdot f(x) < 0$ となる x があるから, 2点 $(x, f(x))$ と $(x-0.001, f(x-0.001))$ を結ぶ直線と x 軸との交点を, 求める実根の近似値とする。なお 0.001 はもっと小さい数の方がよいのであるが, 実行速度を考慮すればこの程度が適当であろう。

(1) 中間値の定理

例 15. $f(x)=x^3-x$ について, $f(-2)=-6$, $f(2)=6$, このとき $f(x)=1$ となる x を求める。

```

100 DEF FNA(X)=X^3-X-1
110 A=-2:B=2
120 FOR X=A TO B STEP .001
130   IF FNA(X)=0 THEN PRINT X:GOTO *EXIT
140   IF FNA(X-.001)*FNA(X)<0 THEN R=.001*FNA(X)/(FNA(X-.001)-FNA(X))+X
150   IF FNA(X-.001)*FNA(X)<0 THEN PRINT R:GOTO *EXIT
160 NEXT X
170 '
180 *EXIT
190 END
    
```

実行結果

R = 1.32471776008606

(参考) print 1.32471776008606^3-1.32471776008606
.999999159190716

(2) 平均値の定理

例 16. $f(x)=x^3-2x$ について $f'(x)=3x^2-2$, このとき $\frac{f(3)-f(1)}{3-1}=f'(x)$ となる x を求める。

```

100 DEF FNA(X)=X^3-2*X:DEF FND(X)=3*X^2-2
110 A=1:B=3
120 DEF FNH(X)=FND(X)-(FNA(B)-FNA(A))/(B-A)
130 FOR X=A TO B STEP .001
    
```

実行結果

R = 2.081665992736816

(参考) print (FNA(3)-FNA(1))/2
11.00000095367432

(以下例 15 と同様)

print FND(2.081665992736816)
11

(3) Cauchy の平均値の定理

例 17. $f(x)=x^3-2x$, $g(x)=4x^2+x$ のとき $f'(x)=3x^2-2$, $g'(x)=8x+1$. このとき $\frac{f(3)-f(1)}{g(3)-g(1)} = \frac{f'(x)}{g'(x)}$ をみたす x を求める。

```

100 DEF FNF(X)=X^3-2*X:DEF FNG(X)=4*X^2+X
110 DEF FNFD(X)=3*X^2-2:DEF FNGD(X)=8*X+1
    
```

```

120 A=1:B=3
130 DEF FNFGD(X)=FNFD(X)/FNGD(X)-(FNF(B)-FNF(A))/(FNG(B)-FNG(A))
140 FOR X=A TO B STEP .001

```

(以下例 15 と同様)

実行結果

R = 2.138159548592676

```

(参考) print (FNF(3)-FNF(1))/(FNG(3)-FNG(1))
        .647058879627901

```

```

print FNFD(2.138159548592676)/FNGD(2.138159548592676)
        .6470588669078637

```

(4) 積分に関する平均値の定理

例 18. $f(x) = \sqrt{1-x^2}$ ($0 \leq x \leq 1$) について $\int_0^1 f(x) dx = f(\xi)$ となる ξ を求める。(積分については例 12 を参照)

実行結果

100 N=250

R = .6190038919448853

110 DEF FNA(X)=SQR(1-X^2)

(参考) print S

120 GOSUB *SIMPSON : ' EXAMPLE 12

.7853878861566385

130 DEF FNB(X)=FNA(X)-S : ' S は積分の値

print SQR(1-0.6190038919448853^2)

(以下例15と同様)

.7853879180106381

§ 4. 行列・行列式

(1) 定義による 4 次行列式の計算

```

480 *CALC4
490 FOR I=1 TO N : FOR J=1 TO N
510   IF J=I THEN GOTO *BRKJ
520   FOR K=1 TO N
530     IF K=I OR K=J THEN GOTO *BRKK
540     FOR L=1 TO N
550       IF L=I OR L=J OR L=K THEN GOTO *BRKL
560       IF (L-I)*(L-J)*(L-K)*(K-I)*(K-J)*(J-I)>0 THEN SG=1 ELSE SG=-1
570       SUM=SUM+SG*A(1,I)*A(2,J)*A(3,K)*A(4,L)
580       *BRKL : NEXT L
590       *BRKK : NEXT K
600       *BRKJ : NEXT J : NEXT I
620 RETURN

```

とすれば, 行列式の定義をそのまま記述したことになる。

RUN

$$\begin{vmatrix} 2 & 5 & -4 & 3 \\ 3 & 2 & 7 & -4 \\ 1 & -6 & 3 & 2 \\ -2 & 3 & 5 & 4 \end{vmatrix} = 2096$$

the time needed = 0

(2) 余因子展開による 4 次行列式の計算

余因子展開のプログラムは多少長めになるし, 速度も定義によるものとそれほど変わらないから, 定義による行列式の計算を行えばよいと思う。

```

410 FOR S=1 TO 4
420   GOSUB *DETERM3 : DETERM=DETERM+(-1)^(N+S)*A(4,S)*AA(S)
440 NEXT S
480 '
560 *DETERM3 : RETURN
RUN

```

$$\begin{vmatrix} 1 & 3 & -2 & 4 \\ 3 & 5 & -7 & 5 \\ 2 & 7 & 1 & 2 \\ -2 & -6 & 7 & -8 \end{vmatrix}$$

= -93

the time needed = 2

(3) 行列の積の計算

プログラムの本体は

```

370 *CALCAB
380   FOR J=1 TO P : FOR K=1 TO R
400     FOR S=1 TO Q
410       C(J,K)=C(J,K)+A(J,S)*B(S,K)
420     NEXT S
430   NEXT K : NEXT J
450 RETURN

```

と簡単であるが、行列を表示するのは多少面倒である。実行結果は

RUN

$$\begin{bmatrix} 125 & 41 & -65 & 34 \\ 51 & -72 & 34 & 64 \\ 33 & 25 & 34 & -48 \end{bmatrix} \begin{bmatrix} 15 & -42 & 18 \\ 343 & -37 & 40 \\ 25 & 47 & -54 \\ -15 & 85 & 26 \end{bmatrix}$$

$$= \begin{bmatrix} 13803 & -6932 & 8284 \\ -24041 & 7560 & -2134 \\ 10640 & -4793 & -1490 \end{bmatrix}$$

(4) 4次行列の余因子行列と逆行列の計算

余因子を求める作業

```

1260 *GETAXY
1280   FOR X=1 TO N : FOR Y=1 TO N : FOR S=1 TO N : FOR T=1 TO N
1320     IF S<X AND T<Y THEN D(S,T)=A(S,T)
1330     IF S<X AND T>Y THEN D(S,T-1)=A(S,T)
1340     IF S>X AND T<Y THEN D(S-1,T)=A(S,T)
1350     IF S>X AND T>Y THEN D(S-1,T-1)=A(S,T)
1360   NEXT T : NEXT S
1370   GOSUB *CALC : AA(X,Y)=(-1)^(X+Y)*DETXY
1390 NEXT Y
1400   DETA=DETA+A(1,X)*AA(1,X)
1410 NEXT X
1420 RETURN

```

を除けば行列の表示が残るだけである。

RUN

Cofactor Matrix

$$\begin{bmatrix} 70 & -52 & -32 & -66 \\ 84 & -66 & -42 & -72 \\ 50 & -32 & -28 & -42 \\ -88 & 70 & 50 & 84 \end{bmatrix}$$

Inverse Matrix

$$\begin{bmatrix} 1 & 2 & 3 & 4 \\ 2 & -3 & 4 & 1 \\ 4 & 1 & -2 & 3 \\ -3 & 4 & 1 & 2 \end{bmatrix}^{-1} = (1/36) \begin{bmatrix} 70 & -52 & -32 & -66 \\ 84 & -66 & -42 & -72 \\ 50 & -32 & -28 & -42 \\ -88 & 70 & 50 & 84 \end{bmatrix}$$

(4) 基本変形・階段行列

行についての基本変形だけで階段行列を作る。整数の範囲で取り扱うための工夫をした。第1列から0でない数を検索し、存在すればその行を第1行と交換しその列の下に位置する行列の要素を0にする。もし存在しなければつぎの列に移る。これにより第1階段に変形できる。同様な操作を繰り返して、 $j \geq r, k \geq r$ なる (j, k) 要素がすべて0になれば階段行列は完成し、行列の階数が判る。

プログラム

```

1360 FOR P=1 TO M
1370   GOSUB *SEARCHPIPOT
1380   IF FLG=0 THEN GOTO *EXIT ELSE GOSUB *EXCHROW : GOSUB *CALC
1400 NEXT P
1410 '
1420 *EXIT : PRINT : PRINT "行列の階数は";AKCNV$(STR$(RANK)) : END

```

で、実行結果は

$$\begin{array}{l} 1 \\ 2 \\ 3 \\ 4 \end{array} \begin{bmatrix} 0 & 0 & 4 & 6 & 8 \\ 0 & 3 & 5 & 7 & 9 \\ 0 & 1 & 3 & 5 & 7 \\ 0 & 0 & 0 & -6 & -12 \end{bmatrix}$$

Ok ? Hit Any-key (Ret-key) :

Devide line 1 by 2

$$\begin{array}{l} 1 \\ 2 \\ 3 \\ 4 \end{array} \begin{bmatrix} 0 & 0 & 2 & 3 & 4 \\ 0 & 3 & 5 & 7 & 9 \\ 0 & 1 & 3 & 5 & 7 \\ 0 & 0 & 0 & -6 & -12 \end{bmatrix}$$

Ok ? Hit Any-key (Ret-key) :

Devide line 4 by 6

$$\begin{array}{l} 1 \\ 2 \\ 3 \\ 4 \end{array} \begin{bmatrix} 0 & 0 & 2 & 3 & 4 \\ 0 & 3 & 5 & 7 & 9 \\ 0 & 1 & 3 & 5 & 7 \\ 0 & 0 & 0 & -1 & -2 \end{bmatrix}$$

Ok ? Hit Any-key (Ret-key) :

Exchange line 1 for line 2

$$\begin{array}{l} 2 \\ 1 \\ 3 \\ 4 \end{array} \begin{bmatrix} 0 & 3 & 5 & 7 & 9 \\ 0 & 0 & 2 & 3 & 4 \\ 0 & 1 & 3 & 5 & 7 \\ 0 & 0 & 0 & -1 & -2 \end{bmatrix}$$

Ok ? Hit Any-key (Ret-key) :

3 ↔ 1

$$\begin{array}{l} 2 \\ 1 \\ 3 \\ 4 \end{array} \begin{bmatrix} 0 & 3 & 5 & 7 & 9 \\ 0 & 0 & 2 & 3 & 4 \\ 0 & 0 & 1 & 2 & 3 \\ 0 & 0 & 0 & -1 & -2 \end{bmatrix}$$

Ok ? Hit Any-key (Ret-key) :

2 ↔ 1

$$\begin{array}{l} 2 \\ 1 \\ 3 \\ 4 \end{array} \left[\begin{array}{cccccc} 0 & 3 & 5 & 7 & 9 \\ 0 & 0 & 2 & 3 & 4 \\ 0 & 0 & 0 & 1 & 2 \\ 0 & 0 & 0 & -1 & -2 \end{array} \right]$$

Ok ? Hit Any-key (Ret-key) :

1 ↔ -1

$$\begin{array}{l} 2 \\ 1 \\ 3 \\ 4 \end{array} \left[\begin{array}{cccccc} 0 & 3 & 5 & 7 & 9 \\ 0 & 0 & 2 & 3 & 4 \\ 0 & 0 & 0 & 1 & 2 \\ 0 & 0 & 0 & 0 & 0 \end{array} \right]$$

Ok ? Hit Any-key (Ret-key) :

行列の階数は 3

(5) 連立1次方程式の解を求める。

係数行列に対して基本変形等の操作を行えばよい。解の表示が面倒であった。実行結果は

$$\left[\begin{array}{cccc} 0 & 0 & 4 & 8 \\ 0 & 0 & 0 & 0 \\ 0 & 4 & 5 & 9 \\ 5 & -2 & -3 & -6 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 7 \\ 0 \\ -6 \\ 8 \end{array} \right]$$

Exchange line 1 for line 4

$$\left[\begin{array}{cccc} 5 & -2 & -3 & -6 \\ 0 & 0 & 0 & 0 \\ 0 & 4 & 5 & 9 \\ 0 & 0 & 4 & 8 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 8 \\ 0 \\ -6 \\ 7 \end{array} \right]$$

Exchange line 2 for line 3

$$\left[\begin{array}{cccc} 5 & -2 & -3 & -6 \\ 0 & 4 & 5 & 9 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 4 & 8 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 8 \\ -6 \\ 0 \\ 7 \end{array} \right]$$

(2,2) ↔ (1,2)

$$\left[\begin{array}{cccc} 10 & 0 & -1 & -3 \\ 0 & 4 & 5 & 9 \\ 0 & 0 & 0 & 0 \\ 0 & 0 & 4 & 8 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 10 \\ -6 \\ 0 \\ 7 \end{array} \right]$$

Exchange line 3 for line 4

$$\left[\begin{array}{cccc} 10 & 0 & -1 & -3 \\ 0 & 4 & 5 & 9 \\ 0 & 0 & 4 & 8 \\ 0 & 0 & 0 & 0 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 10 \\ -6 \\ 7 \\ 0 \end{array} \right]$$

(3,3) ↔ (1,3)

$$\left[\begin{array}{cccc} 40 & 0 & 0 & -4 \\ 0 & 4 & 5 & 9 \\ 0 & 0 & 4 & 8 \\ 0 & 0 & 0 & 0 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 47 \\ -6 \\ 7 \\ 0 \end{array} \right]$$

(3,3) ↔ (2,3)

$$\left[\begin{array}{cccc} 40 & 0 & 0 & -4 \\ 0 & 16 & 0 & -4 \\ 0 & 0 & 4 & 8 \\ 0 & 0 & 0 & 0 \end{array} \right] \left[\begin{array}{c} X1 \\ X2 \\ X3 \\ X4 \end{array} \right] = \left[\begin{array}{c} 47 \\ -59 \\ 7 \\ 0 \end{array} \right]$$

Independent variables : X4

$$X1 = (47/40) + (1/10) * t$$

$$X2 = (-59/16) + (1/4) * t$$

$$X3 = (7/4) + (-2) * t$$

$$X4 = (0) + (1) * t$$

(6) プログラムの例

ここで、われわれの用いたプログラムの中から代表として連立1次方程式の解を求めるものを示す。

```

1000 ' prog16.BAS",A  :' 連立1次方程式の解
1010 '
1020 CLS : CONSOLE 0,24,0,1
1030 N=4 : DIM A(N,N+1),C(N+1),CA$(N,N+1),NN$(N),SOL$(N)
1040 '
1050 DATA 0, 0, 4, 8, 7
1060 DATA 0, 0, 0, 0, 0
1070 DATA 0, 4, 5, 9, -6
1080 DATA 5, -2, -3, -6, 8
1090 '
1100 DATA 1, 1, 1, 3, 3
1110 DATA 1, -1, -5, -1, -5
1120 DATA 3, 1, -3, 5, 1
1130 DATA 5, 1, -7, 7, -1
1140 '
1150 DATA 1, 2, 3, 2, 3
1160 DATA 2, 4, 6, 3, 2
1170 DATA 3, 6, 13, 8, 2
1180 DATA 4, 8, 8, 7, 23
1190 '
1200 DATA 0, 0, 0, 0, 0
1210 DATA 0, 0, 0, 0, 0
1220 DATA 0, 0, 0, 0, 0
1230 DATA 0, 0, 0, 0, 0
1240 '
1250 RESTORE 1050  :' DATA を変えるときはこの行の番号を変える。
1260 FOR J=1 TO N : FOR K=1 TO N+1 : READ JK : A(J,K)=JK : NEXT K : NEXT J
1270 GOSUB *DISPLAY
1280 FOR JK=1 TO N                                     :' check common divisor
1290   FOR KJ=1 TO N+1 : C(KJ)=A(JK,KJ) : NEXT KJ : GOSUB *GCDARREY
1300   FOR KJ=1 TO N+1 : A(JK,KJ)=C(KJ) : NEXT KJ
1310   IF GCD<>1 THEN COLOR 6 : PRINT "Devide line";JK;"by";GCD :
                                     COLOR 7 : GOSUB *DISPLAY
1320 NEXT JK
1330 '
1340 ' ----- Main Routine
1350 FOR P=1 TO N
1360   GOSUB *SEARCHNONZERO
1370   IF FLGC=N+1 THEN PRINT "解なし" : END
1380   IF FLG=0 THEN GOTO *EXIT
1390   GOSUB *EXCHANGE : GOSUB *ELEMENTTRANSFORM
1400 NEXT P
1410 ' -----
1420 *EXIT : GOSUB *DISPSOLUTION : END
1430 '

```

```

1440 *SEARCHNONZERO                                :' 検索
1450   FLG=0 : RANK=0
1460   FOR K=P TO N+1
1470     FOR J=P TO N
1480       IF A(J,K)<>0 THEN FLGR=J:FLGC=K:FLG=1:RANK=RANK+1: GOTO *BRK1
1490     NEXT J
1500   NEXT K
1510 *BRK1
1520 RETURN
1530 '
1540 *EXCHANGE                                      :' 行の交換
1550   FOR K=1 TO N+1
1560     SWAP A(P,K),A(FLGR,K)
1570   NEXT K
1580   IF A(P,FLGC)<>A(FLGR,FLGC) THEN COLOR 6 : PRINT "Exchange line";P;
                                     "for line";FLGR : COLOR 7 : GOSUB *DISPLAY
1590 RETURN
1600 '
1610 *ELEMENTTRANSFORM                            :' 基本変形
1620   FOR R=1 TO N
1630     IF R=P THEN GOTO *BRK2
1640     AA=A(P,FLGC) : BB=A(R,FLGC)
1650     IF BB=0 THEN GOTO *BRK2
1660     COLOR 6 : PRINT " (";P;",";FLGC;") <--> (";R;",";FLGC;")" : COLOR 7
1670     GOSUB *GCDTWO
1680     FOR K=1 TO N+1 : A(R,K)=AA*A(R,K)-BB*A(P,K) : NEXT K
1690     FOR K=1 TO N+1 : C(K)=A(R,K) : NEXT K
1700     GOSUB *GCDARREY
1710     FOR K=1 TO N+1 : A(R,K)=C(K) : NEXT K
1720     GOSUB *DISPLAY
1730   *BRK2
1740 NEXT R
1750 RETURN
1760 '
1770 *GCDTWO                                         :' 2数の簡約
1780   IF AA=0 OR BB=0 THEN GCD=1 : GOTO *BRK3
1790   GAA=ABS(AA) : GBB=ABS(BB)
1800   IF GAA<=GBB THEN SWAP GAA,GBB
1810 *LOOPF : GRR=GAA-(GAA \ GBB)*GBB
1820   IF GRR=0 THEN GCD=GBB : GOTO *BRK3
1830   IF GRR<>0 THEN GAA=GBB:GBB=GRR : GOTO *LOOPF
1840 *BRK3
1850   IF GCD=0 THEN GCD=1
1860   BB=BB/GCD : AA=AA/GCD
1870 RETURN

```

```

1880 '
1890 *GCDARREY                                     : ' 行を簡約
1900   GCD=0
1910   FOR J=1 TO N+1
1920     IF C(J)<>0 THEN GCD=C(J) : GOTO *BRK4
1930   NEXT J
1940 *BRK4
1950   IF GCD=0 THEN GCD=1 : RETURN
1960   FOR K=1 TO N+1
1970     GAA=GCD:GBB=ABS(C(K))
1980     IF GBB=0 THEN GCD=GAA : GOTO *BRK5
1990     IF GAA<=GBB THEN SWAP GAA,GBB
2000     *LOOPS : GRR=GAA-(GAA \ GBB)*GBB
2010     IF GRR=0 THEN GCD=GBB : GOTO *BRK5
2020     IF GRR<>0 THEN GAA=GBB:GBB=GRR : GOTO *LOOPS
2030     *BRK5
2040   NEXT K
2050   FOR K=1 TO N+1 : C(K)=C(K)/GCD : NEXT K
2060   RETURN
2070 '
2080 *DISPSOLUTION                                   : ' 解の表現
2090   FOR J=1 TO N
2100     FOR K=1 TO N+1
2110       CA$(J,K)=RIGHT$(STR$(A(J,K)),LEN(STR$(A(J,K))))
2120     NEXT K
2130   NEXT J
2140   FOR K=1 TO N
2150     CA$(0,K)=RIGHT$(STR$(K),1) : A(0,K)=K
2160   NEXT K
2170   FOR S=1 TO RANK
2180     FOR T=1 TO N
2190       IF A(S,T)<>0 THEN PPV=T : GOTO *BRK6
2200     NEXT T
2210     *BRK6
2220     FOR K=PPV TO S+1 STEP -1
2230       FOR J=0 TO N
2240         SWAP CA$(J,K),CA$(J,K-1) : SWAP A(J,K),A(J,K-1)
2250       NEXT J
2260     NEXT K
2270   NEXT S
2280   COLOR 6 : PRINT " Independent variables : " ;; COLOR 7
2290   FOR K=RANK+1 TO N
2300     IF K<N THEN COLOR 5 : PRINT " X";CA$(0,K);" , " ;; COLOR 7
2310     IF K=N THEN COLOR 5 : PRINT " X";CA$(0,K) ;; COLOR 7
2320   NEXT K

```

```

2330 PRINT
2340 FOR J=1 TO N
2350   AA=A(J,J):BB=A(J,N+1) : GOSUB *SOLUTION : CA$(J,N+1)=C$
2360   FOR K=1 TO N
2370     AA=A(J,J):BB=-A(J,K) : GOSUB *SOLUTION: CA$(J,K)=C$
2380   NEXT K
2390 NEXT J
2400 FOR J=1 TO N
2410   GOSUB *INDEPVARIBLE
2420   SOL$(J)="X"+CA$(0,J)+"= (" +CA$(J,N+1)+")"+SUM$
2430 NEXT J
2440 PRINT
2450 FOR K=1 TO N           : ' FOR K=1 TO N : PRINT SOL$(K) : NEXT K
2460   FOR J=1 TO N
2470     IF K=A(0,J) THEN PRINT SOL$(J)
2480   NEXT J
2490 NEXT K
2500 PRINT
2510 RETURN
2520 '
2530 *INDEPVARIBLE
2540 SUM$=""
2550 FOR K=RANK+1 TO N
2560   IF J=K THEN CA$(J,K)="      1 "
2570   SUM$=SUM$+" + (" +CA$(J,K)+")*t"+NN$(K-RANK)
2580 NEXT K
2590 RETURN
2600 '
2610 *SOLUTION
2620 GOSUB *GCDTWO
2630 AA$=STR$(AA) : BB$=STR$(BB)
2640 AA$=RIGHT$(AA$,LEN(AA$)-1) : BB$=RIGHT$(BB$,LEN(BB$)-1)
2650 IF AA*BB>0 AND (AA<>1 OR AA<>-1) THEN C$=BB$+"/"+AA$
2660 IF AA*BB>0 AND (AA=1 OR AA=-1) THEN C$=BB$+" "
2670 IF AA*BB=0 THEN C$=" 0 "
2680 IF AA*BB<0 AND (AA<>1 OR AA<>-1) THEN C$="-"+BB$+"/"+AA$
2690 IF AA*BB<0 AND (AA=1 OR AA=-1) THEN C$="-"+BB$+" "
2700 C$=RIGHT$("          "+C$,8)
2710 RETURN
2720 '
2730 *DISPLAY                                     : ' データの表現
2740 PRINT "  ";SPC(7*N);"  ";SPC(2);"  ";SPC(7);"  "
2750 FOR J=1 TO N
2760   PRINT " | " ;
2770   FOR K=1 TO N

```

```

2780     PRINT RIGHT$("      "+STR$(A(J,K)),7) ;
2790     NEXT K
2800     PRINT " | ";
2810     PRINT " | X";RIGHT$(" "+STR$(J),1);" | " ;
2820     IF J=INT((N+1)/2) THEN PRINT "= | ";RIGHT$("      "+STR$(A(J,N+1)),7);
        " | " ELSE PRINT " | ";RIGHT$("      "+STR$(A(J,N+1)),7);" | "
2830 NEXT J
2840 PRINT "└";SPC(7*N);"┐└";SPC(2);"┐└";SPC(7);"┐┐"
2850 INN$=INPUT$(1) : PRINT
2860 *WAITING : IF INN$="" THEN GOTO *WAITING
2870 RETURN

```

参 考 文 献

- 1) 木田祐司 1989. 多倍長計算用BASIC,UBASIC86 Ver. 7.
- 2) 扇谷 尚 昭和56年5月. 高等学校学習指導要領改訂にかかわる大学教育の問題. 一般教育学会誌 第3巻 第1号.
- 3) 尾崎康弘 昭和62年5月. 数学教育へのパソコン導入の試み. 一般教育学会誌 第9巻 第1号.
- 4) 中学生の数学成績と教師の指導法. 昭和58年3月. 国立教育研究所紀要 第107集 第2回国際数学教育調査国内報告.
- 5) 馬野元生・西川泰行・濱谷英次 1985. 高校数学とコンピュータ. 別冊・数学セミナー「コンピュータと数学」: 122-132, 日本評論社.
- 6) You And Your Students, Prepared by a Faculty Committee at the Massachusetts Institute of Technology. IDE 教育資料 第44集. 1980.

Summary

We considered the practical use of a personal computer in mathematics in the general education at college level. As a result, we found that in mathematics in general education at college level, computers are effective for numerical calculations and drawing graphs.

We understand that we can apply computers in mathematical education for general education at college level in the future.